



Simple UART Intellectual Property - VHDL

Introduction

A few years ago, we have developed and used a simple yet efficient UART module (both in VHDL and Verilog, though this document only describes the VHDL version). Designing this core of the UART was a rather simple task of no more than a few hours, and it performed very well where we needed it. For simulation purpose, we did also develop a behavioral (i.e. not synthesizable) model.

In some conferences (which you may find at <http://www.alse-fr.com>), we used this project as an example to demonstrate how HDL Design techniques (and especially Finite State Machines) can solve apparently complex problems with a minimal effort.

Many years after, we are still asked to provide the complete source code. Apparently, despite the simplicity of the project and the availability of several free models, there is still a demand for it.

Therefore, we took the decision to clean it a bit, add (and remove) a few features, comment the code, enhance the test bench, create a working example, and make it available. This is what you have in your hands now. We designed it also as a design and methodology example. Beginners will probably learn a lot by understanding any detail of the RTL, the simulation, and the behavioral pieces of code.

Experts have undoubtedly stopped reading this document and have jumped to the code ☺.

If you plan to use this IP for anything else than a purely educational project, contact ALSE and purchase our very cheap UART Kit that includes this IP and a Versatile version (dynamically configurable, and support for various data formats and parity).

Main Features

- Asynchronous RS232 character-based transmit / receive function
- Very compact (around 50 CLB slices in the Spartan II family, or less than 100 LEs in Altera technologies). This includes the baud rates generators...
- Good timing performance. Except in very trivial designs, the UART won't appear in the critical paths.
- No internal Fifo. The user application is supposed to handle the communication latencies and to consume received characters faster than they are received (very general case). If needed, the characters could be pipelined in an external Fifo.
- Internal Baud rate generators included (fixed value, by generic parameter in this simple version).
- Synthesizable to any technology (FPGA, CPLD, or Asic -not allowed by the agreement-).
- Code tested on most Synthesis tools available today. Synthesis and P&R Script files provided for Leonardo Spectrum, Synplify, Quartus II and ISE.
- Demo application to test the UART on most of the existing FPGA demo boards.
- Simulation test bench including reusable behavioral UART and RS232 Terminal (console) simulation with File I/O and results logged in the transcript. A complete simulation script is also provided.

Note that this core is not designed as a computer peripheral (or SoC equivalent). It performs best in a hardware design context, where no software (i.e. no CPU) is involved.

Archive Contents

The archive is organized as follows:

Directory	Contents
Doc	PDF documentation (you're reading it now)
Src	RTL (synthesizable) VHDL Source files
Simu	Test bench, behavioral model, Simulation scripts & I/O file
Fit	Place and Route (FPGA project) base directory.
Fit / Ise	Ready-to-use Xilinx-ISE project for Insight Spartan II board.
Fit / Ise_Tcl	Ready-to-use Xilinx-ISE "does-all" script.
Fit / Tornado	Ready-to-use scripts for Tornado board.
Fit / Nios1C20	Use this if your target is an Altera NIOS 1C20 board

Note that each subdirectory has a small readme.txt file.

Where to go from here ?

Indeed, different users will handle this project in many different ways, so we consider that if you're reading this chapter, you are probably expecting a little help. Once the archive has been unpacked in an empty directory (with no space in the path name), you might consider the following steps:

1. Inspect **UARTS.vhd**. This is the RTL code that likely brought you here. Make sure you understand completely how it works. The way this UART is implemented (Finite State Machines) is a very powerful methodology, which is applicable to most designs of all kinds.
You'll notice that a reasonable use of generic parameters, functions, constants, etc... can help you create a very reusable yet efficient RTL code.
2. Examine **APPLIC.vhd**. This one is really simple. Its task is to establish a dialog with UARTS to (modify and) echo the characters received. No surprise: we used an FSM here too.
3. Look at **TOP_UART.vhd**. Do not modify it.
4. Open **CONSOLE.vhd**. This is a simulation-only file that reads in a text file and creates an RS232 stream from it. The text file contains simulation commands (delay = '#') and comments.
"Console" includes a behavioral RS232 transmitter, which can be reused in other situations where it is necessary to send RS232 characters, strings, frames, commands, text files...
5. Open **TB_TOP.vhd**. It includes a behavioral RS232 receiver with transcript logging, which can be reused in other projects.
6. **Simulate the whole design**. You may examine the proposed signals in the waveform to better understand how UARTS works. Don't forget to look at the transcript...
7. **Adapt TOP_UART.vhd** to your hardware (FPGA board) :
 - * modify the Generics default value (**entity** section) for the System clock frequency.
 - * modify the **UARTS Generic map** section (Baud1 & Baud2).
 - * modify the **Reset polarity** if necessary.
 - * you may force DIPSW to a fixed level if you don't want to keep two different baud rates,
 - * etc...
8. **Synthesize**, do the **pin assignment, place & route** your project.
9. **Download** the project, connect a PC and test the communication.
Don't hesitate to use an oscilloscope to verify the main signals and their polarities...


UARTS

This is the UART Synthesizable (RTL) module, “*the IP*” and we will describe its parameters, inputs and outputs. In a following section, we will describe a typical application using this module.

Note that we did our best to keep this module as simple as possible.

Generics:

Fxtal	System clock Freq in Hertz.
Baud	Baud Rate
<i>Parity</i>	<i>Removed in the Simple version</i>
<i>Even</i>	<i>Removed in the Simple version</i>
<i>Baud2</i>	<i>Removed in the Simple version</i>



Inputs / Outputs:

CLK	System Clock, frequency must be declared in the Generic map.
RST	Asynchronous active-high reset.
<i>Baud</i>	<i>Removed in the Simple version (Toggles between Baud1 & Baud2 rates)</i>
Din[7:0]	Eight-bits Data : raw character to be sent.
LD	Load pulse (one clock cycle) to load Din in the Transmitter and debut the transmission immediately. The application is supposed to verify beforehand that TxBusy is false and that the external device is ready to receive the character (flow control / RTS).
Rx	RS232 receive signal input. Is '1' when the line is idle.
Tx	RS232 transmit signal output. Is '1' when the line is idle.
TxBusy	A '1' indicates that the UART is busy sending the previous character and will ignore a LD request (the character would be lost).
Dout[7:0]	Data received. Read this value when the RxRDY pin is '1'
RxRDY	A '1' pulse (one system clock cycle long) indicates that a character is received and is available at Dout.
RxErr	A '1' here indicates that a character was received incorrectly. In this case, no RxRDY would occur.

Note that the description is smart enough to size automatically the baud rate generator's internal counters. However, the user must verify that the combination of system clock frequency and Baud Rate selected are compatible in terms of **timing inaccuracies**.

Example: Fxtal = 10 MHz, Baud=115200 => Bit period = 8.68 μ s

The **Half bit period** generation requires a division by : 43.4 which is rounded to 43. The equivalent Baud rate would then be $1e7 / 86 = \sim 116.300$, i.e. an error of $\sim 1\%$, which is in this case acceptable :

Beware: this error does accumulate ! A 1 % error leads to $\sim 20\%$ error at the last (stop) bit... still acceptable but probably near the limit for safe communication.

Example Design

To demonstrate the use of the UARTS module, we have built a small application that reads in characters from the RS232 line, and echoes them after modification. When a question mark is sent, the system replies by a prompt line saying "Scrambler Demo."

This design can be completely simulated. It can also be fitted on a real FPGA board to test the implementation, the RS232 connection, etc...

A PC running HyperTerminal or uCon (eg) can be then used to exercise the design manually.

The module name "APPLIC" does handle the UART module completely :


- Waits until a character is received by UARTS.
- If this character is Cr, Lf, or space, it is echoed intact.
- If the character is a digit ('0' .. '9'), its value is memorized and a dot '.' is echoed
- Otherwise, the Ascii code is incremented by the memorized value and the new character is echoed.

Note: if for some reason the outside system is busy, the character to be echoed may be lost.

The example design's top level is *roughly* equivalent to the following schematics:

UARTS : typical application

This can be tested on an FPGA demo board.



ALSE
 info@alse-fr.com
 http://www.alse-fr.com

Title: TOP

Name:

Date: 3/23/03

Sheet 1 of 1

Note : this schematics, when converted by the recent versions of Xilinx ISE gets ruined :-(
 So, for the ISE projects, we rely on a VHDL top-level (next page) rather than schematics (and we use a Tcl script so running ISE isn't even necessary).

```

-----
--   UART-based Demo Project : Top Level
-----
-- (c) A.L.S.E.
-- Version : 3.1 - simplified
-- Author   : Bert Cuzeau
-- E_mail   : info@alse-fr.com
-- Web      : http://www.alse-fr.com
--
-- Note : can be used in any existing demo board like
-- Tornado, Altera NIOS, Insight Spartan2, Spartan3, Xess, Actel, Lattice, etc....
-- How to use :
-- * modify the Generic Fxtal default value to the clock input Frequency
-- * modify the Generic Baud default to the desired baud rate.
-- * Assign the pins, build the FPGA project, and run !
Library IEEE;
use IEEE.std_logic_1164.ALL;
-----
Entity TOP_UART is
-----
-- >>>> IMPORTANT : modify Fxtal, Baud & Reset polarity below to match your board !!!!
Generic ( Fxtal      : integer := 50E6; -- CLK requery, in Hertz (60Mhz=60E6)
          Baud       : integer range 1200 to 921600 := 115200; -- Baud rate
          Rst_Pol    : std_logic := '0' ); -- Reset polarity ('0' means active Low)
Port (     CLK      : In      std_logic; -- Global Clock
          Reset     : In      std_logic; -- Async reset, active @ Rst_Pol !
          RS232_RX  : In      std_logic; -- RS232 RX input
          RS232_TX  : Out     std_logic -- RS232 TX output );
end TOP_UART;
-----
Architecture Schematics of TOP_UART is
-----
signal RSTi : std_logic; -- active High re-synchronized reset
signal RXErr : STD_LOGIC;
signal RXRDY : STD_LOGIC;
signal TxBusy : STD_LOGIC;
signal LD_SDout : STD_LOGIC;
signal SDout : STD_LOGIC_VECTOR (7 downto 0);
signal SDin : STD_LOGIC_VECTOR (7 downto 0);
signal SDCoded : STD_LOGIC_VECTOR (7 downto 0);
begin

process (CLK) -- Global Reset resynchronized
begin
if rising_edge(CLK) then
RSTi <= Reset xor not Rst_Pol;
end if;
end process;

-- (direct) Instanciations
i_APPLIC : entity work.APPLIC
Port Map ( CLK => CLK,
           RST => RSTi,
           RXErr => RXErr,
           RXRDY => RXRDY,
           SDin => SDin,
           TxBusy => TxBusy,
           SDout => SDout,
           LD_SDout=> LD_SDout );

i_UARTS : entity work.UARTS
Generic map ( Fxtal => Fxtal,
             Baud => Baud )
Port Map ( CLK => CLK,
           RST => RSTi,
           Din => SDout,
           LD => LD_SDout,
           RX => RS232_RX,
           Dout => SDin,
           RXErr => RXErr,
           RXRDY => RXRDY,
           TX => RS232_TX,
           TxBusy => TxBusy );
end Schematics;

```

APPLIC Code

Let us review the architecture code for APPLIC:

```
-----  
Architecture RTL of APPLIC is  
-----  
  
-- Next function commented out for tools that don't support Character'pos !  
-- (Quartus II does support this nicely)  
-- function Char2SLV8 (c : character) return std_logic_vector is  
-- begin return std_logic_vector(to_unsigned(character'pos(c),8));  
-- end function;  
  
constant Prompt : string := "Scrambler Demo." & Cr & Lf;  
subtype Byte is std_logic_vector (7 downto 0);  
type Hex_string is array (natural range <>) of Byte;  
  
-- For tools that do not support Character'pos ! (Xilinx) :  
constant Prompt_hex : Hex_string  
:= ( x"53", x"63", x"72", x"61", x"6D", x"62", x"6C", x"65",  
     x"72", x"20", x"44", x"65", x"6D", x"6F", x"2E", x"0D",  
     x"0A" );  
  
type State_Type is (Boot1, Boot2, Boot3, Idle, Scramble, Send);  
signal State : State_Type;  
  
signal RData : std_logic_vector (7 downto 0);  
signal SData : std_logic_vector (7 downto 0);  
  
signal Code : unsigned(3 downto 0);  
signal CharCount : natural range Prompt_hex'low to Prompt_hex'high+1;  
signal CharAvail : boolean;  
  
begin  
SDout <= SData;  
  
process (CLK,RST)  
begin  
  if RST='1' then  
    State <= Boot1;  
    LD_SDout <= '0';  
    SData <= (others=>'0');  
    RData <= (others=>'0');  
    Code <= (others=>'0');  
    CharCount <= Prompt_hex'low;  
    CharAvail <= false;  
  
  elsif rising_edge(CLK) then  
  
    -- memorize any arriving character  
    if RXRDY='1' then -- memorize the incoming char  
      RData <= SDin;  
      CharAvail <= true;  
    end if;  
  
    LD_SDout <= '0';  
  
    case State is  
  
      when Boot1 =>  
        CharCount <= Prompt_hex'low;  
        State <= Boot2;  
  
      when Boot2 =>  
        if (TxBusy='0') then  
          -- SData <= Char2SLV8 (Prompt(CharCount));  
          SData <= Prompt_hex(CharCount);  
          LD_SDout <= '1';  
          CharCount <= CharCount+1;  
  

```

```

        if CharCount=Prompt_hex'high then
            State <= Idle;
        else
            State <= Boot3;
        end if;
    end if;

    when Boot3 => -- in case we need one clock cycle before testing Busy
        State <= Boot2;

    when Idle =>
        if CharAvail then
            CharAvail <= false;
            State <= Scramble;
        end if;

    when Scramble =>
        if (TxBusy='0') then
            State <= Send;
            case to_integer(unsigned(RData)) is
                when 10|13|32 => -- do not translate Cr Lf Space !!!
                    SData <= RData;
                when 48 to 57 => -- change the offset & returns a dot
                    Code <= unsigned(RData(3 downto 0)); -- x"30"..x"3F"
                    SData <= x"2E"; -- Char2SLV8('.');
                when 63 => -- '?' -> Prompt
                    State <= Boot1; -- override Send
                when others => -- add the current offset
                    SData <= std_logic_vector(unsigned(RData) + Code);
            end case;
        end if;

    when Send =>
        LD_SDout <= '1';
        State <= Idle;

    end case;

end if;
end process;

end RTL;

```

APPLIC is designed using a **Resynchronized Mealy Finite State Machine**, a style we like since it's easy to write, understand and maintain, and less error-prone than combinational process-based styles.

Note the function Char2SLV8, which converts a character into an Std_logic_vector (7 downto 0). This function was not accepted by some synthesis tools (Xilinx) so it is commented out. If your tool supports Character'pos and functions (it should !), you can comment it out and use the ASCII string instead of the hexadecimal array !

Important : the duration of the **RxRdy** signal from UARTS is **only one clock cycle !**
It is your (application's) responsibility to capture this event (and the associated Data word).

Typically a polled peripheral (as in a CPU environment) has to provide a handshake mechanism: a FlipFlop is synchronously set by RxRdy and cleared when the Data is used. This is precisely what **CharAvail** does implement ! You *must* provide a similar feature in your own application.

The Scramble + Send states represent the input character processing (translated and echoed back, basically). Note that the "?" character triggers the sending (response) of the prompt line.

You might find handy to reuse APPLIC as a skeleton for your own processing.

A possible application could be an Intel-Hex format reader, which would fill a memory with the binary contents, and issue an Ack / Nack response to the sender if the line checksum is correct or not. We have already done this for a couple of designs. Contact us if you have a similar need. Actually, writing the VHDL code for the Intel Hex reader is as simple as a software solution, and it executes a lot faster !

Functional Simulation

The `./Simu` sub-directory includes a script named “`simu.do`” which does create the simulation library, compile all the source files, brings interesting signals in the waveform window, and runs the simulation.

The test bench does analyze the RS232 line, and logs the characters echoed in the simulation transcript. Therefore, examining the waveform window is only useful to understand how the system works internally. The characters echoed are printed in the transcript under the form :

```
# === Simulation starting now ===
#
# Character received (hex) = 53 - 'S'
# Character received (hex) = 63 - 'c'
# Character received (hex) = 72 - 'r'
# Character received (hex) = 61 - 'a'
# Character received (hex) = 6D - 'm'
# Character received (hex) = 62 - 'b'
# Character received (hex) = 6C - 'l'
# Character received (hex) = 65 - 'e'
# Character received (hex) = 72 - 'r'
# Character received (hex) = 20 - ' '
# Character received (hex) = 44 - 'D'
# Character received (hex) = 65 - 'e'
# Character received (hex) = 6D - 'm'
# Character received (hex) = 6F - 'o'
# Character received (hex) = 2E - '.'
# Character received (hex) = 0D - '\n'
#
# Character received (hex) = 0A - '\n'
#
# Character received (hex) = 2E - '.'
# Character received (hex) = 48 - 'H'
# Character received (hex) = 2E - '.'
# Character received (hex) = 65 - 'e'
# Character received (hex) = 2E - '.'
# Character received (hex) = 6C - 'l'
# Character received (hex) = 2E - '.'
# Character received (hex) = 6C - 'l'
# Character received (hex) = 2E - '.'
# Character received (hex) = 6F - 'o'
# Character received (hex) = 20 - ' '
# Character received (hex) = 2E - '.'
# Character received (hex) = 57 - 'w'
# Character received (hex) = 2E - '.'
# Character received (hex) = 6F - 'o'
# Character received (hex) = 2E - '.'
# Character received (hex) = 72 - 'r'
# Character received (hex) = 2E - '.'
# Character received (hex) = 6C - 'l'
# Character received (hex) = 2E - '.'
# Character received (hex) = 64 - 'd'
#
# Character received (hex) = 53 - 'S'
# Character received (hex) = 63 - 'c'
# Character received (hex) = 72 - 'r'
# Character received (hex) = 61 - 'a'
# Character received (hex) = 6D - 'm'
# Character received (hex) = 62 - 'b'
# Character received (hex) = 6C - 'l'
# Character received (hex) = 65 - 'e'
# Character received (hex) = 72 - 'r'
#
# etc..
#
# ** Note: End of Simulation.
```

Here is the ASCII script file, named **console.txt**, which generated the transcript above:

```
; -- Console.txt : RSR 232 Input File ---
; Characters are sent as such, except for :
; - semicolons ';' which start a comment ending
;   at the end of the line
; - '#' followed by a delay expressed in micro-seconds.
; Note that Cr & LF are not sent.
;
;#2000
;0H1d2j3i4k
;#80
; 5R6i7k8d9[
;#200
;?
;#2000
```

The initial 2 ms delay leaves room for the Prompt line (which is sent after the system reset).

The seemingly gibberish "0H1d2j3i4k5R6i7k8d9[" is in fact "Hello World", with variable offsets.

The question mark does also re-send the prompt string, hence the last (2 ms) delay.

When the end of the text file is reached, the simulation stops after a small delay.

Synthesis

Before this step, you need first to set up the top level to match your actual hardware & configuration. Open the VHDL file **TOP_UART.vhd** and set the following parameters in the entity generics declaration :

- **System Clock Frequency** as default value of **Fxtal**,
- Desired **Baud rate** as default value for **Baud**,
- **Reset active level** as default value of **Reset_active**,

*Note : Word format with **Parity** yes/no + **Even/Odd**, have been removed in the Simplified version.*

The synthesis project consists of three VHDL files :

UARTS.vhd + **APPLIC.vhd** + **TOP_UART.vhd**, the latter being the top level.

If you use **ISE** version 9.2 or greater, you'll find a ready-to-use script file in `./Fit/lse_tcl` which can be modified easily to match any Xilinx board. Double clicking on "make.bat" builds everything automatically (including the programming file) without having to run ISE ! Do not forget to adapt the UCF constraint file for correct pin assignment.

If you use **Quartus II**, you'll find Tcl scripts for the NIOS 1C20 board and **Tornado** board. Double clicking on "make.bat" builds everything automatically and even download the Tornado board.

All the scripts we provide are relocatable and should work wherever you decide to unpack the archive. However, it is strongly recommended to **avoid path names including space character(s)** (PC).

A note about some Synthesis tools (Warp)

Not all VHDL synthesis tools are equal... Some old obsolete tools like Cypress' Warp do not support the language very well. Some other tools may have trouble with some constructs. In this case, we have rewritten a simplified version of the code under the names **xxx_warp.***. Contact us if you would like to receive this version.

Testing

Prepare the board by attaching a serial cable to the PC's serial line connector (or to an RS232 terminal).

If you use a PC, run Hyperterminal (for example) and adjust the communication parameters to the project's parameters. We suggest 115200 / N / 8 / 1 for Tornado, or 19200 / N / 8 / 1 for an Insight Spartan2 board (which RS232 level translator is a bit lazy). *This must indeed match your generic parameters !*

Last minute note : Microsoft Vista seems to not have Hyperterminal any more ! Yet another reason to stay away from Vista if you can... If you insist on using Vista, install Hyperterminal or another serial terminal emulator like the free MicroCon (uCon) from Ed Sutter (search google for "ucon sutter").

Once the FPGA has been placed and routed as described above, you should be able to download the bitstream into the FPGA. Once this is done, a LED may blink on some boards whenever a character is received from the terminal. If this is not the case, verify that the cable is of the correct type and that the communication software is tuned to the proper COM port.

Type "ABC1DEF". The echo should be "ABC.EFG".

If this doesn't work immediately, then the DIPswitch is probably in a wrong position, or Hyperterminal isn't set correctly...

Note : for the Nios 1C20 board, use the "debug" port (or modify the Tcl script so the pins assigned use the "Console" port).

Some users have used baud rates up to nearly 1 Mega bauds successfully. The trick is to use a USB <-> Serial adapter and hook to the 5 Volts or 3.3 Volts Rx and Tx signals directly, bypassing the RS232 level translators (on both ends). At 50 slices (or 100 Les) the UART and just one or two wires, this becomes a very neat and inexpensive debugging addition for a design which had no provision for this.

On the PC side, we like to use Tcl/Tk as a means to create nice graphical panels exchanging data over the RS232 port with the FPGA.

Conclusion

Besides being a robust, compact and efficient UART, this simple IP should also be a solid and valuable source of information for beginners and even for more experienced designers. Very experienced designers and real experts would have their own code already and wouldn't be using this IP anyway.

The coding style used is directly applicable to all kinds of similar functions, like PS/2 (available from our Website), SPI etc... The simulation models and techniques included are also of great value.

Last but not least, the Tcl scripts for synthesis may also be reused with profit for all projects.

We hope that you will enjoy this UART and that you'll appreciate the sharing of our know-how !

As a return, we will enjoy hearing from your application and experience.

Bert Cuzeau
Paris – Oct 2007
info at alse-fr.com

--oOo--